Rules for integrands involving gamma functions

1. $\int u\,\text{Gamma}[n,\ a+b\,x]\,dx$

1: $\int \text{Gamma}[n,\ a+b\,x]\,dx$

Derivation: Integration by parts

Basis: $\partial_x \text{Gamma}[n,\ a+b\,x] == -\frac{b\,(a+b\,x)^{n-1}}{e^{a+b\,x}}$

Rule:

$$\int \text{Gamma}[n,\ a+b\,x]\,dx \longrightarrow \frac{(a+b\,x)\,\text{Gamma}[n,\ a+b\,x]}{b} + \int \frac{(a+b\,x)^n}{e^{a+b\,x}}\,dx \longrightarrow \frac{(a+b\,x)\,\text{Gamma}[n,\ a+b\,x]}{b} - \frac{\text{Gamma}[n+1,\ a+b\,x]}{b}$$

Program code:

```
Int[Gamma[n_,a_.+b_.*x_],x_Symbol] :=
  (a+b*x)*Gamma[n,a+b*x]/b - Gamma[n+1,a+b*x]/b /;
FreeQ[{a,b,n},x]
```

2. $\int (d\,x)^m\,\text{Gamma}[n,\ b\,x]\,dx$

1. $\int \frac{\text{Gamma}[n,\ b\,x]}{x}\,dx$

1. $\int \frac{\text{Gamma}[n,\ b\,x]}{x}\,dx$ when $n \in \mathbb{Z}$

1: $\int \frac{\text{Gamma}[0,\ b\,x]}{x}\,dx$

Basis: $\text{Gamma}[0,\ z] == \text{ExpIntegralE}[1,\ z]$

Rule:

$$\int \frac{\text{Gamma}[0,\ b\,x]}{x}\,dx \longrightarrow \int \frac{\text{ExpIntegralE}[1,\ b\,x]}{x}\,dx$$

$$\longrightarrow b\,x\,\text{HypergeometricPFQ}[\{1,1,1\},\{2,2,2\},-b\,x] - \text{EulerGamma}\,\text{Log}[x] - \frac{1}{2}\,\text{Log}[b\,x]^2$$

Program code:

```
Int[Gamma[0,b_.*x_]/x_,x_Symbol] :=
  b*x*HypergeometricPFQ[{1,1,1},{2,2,2},-b*x] - EulerGamma*Log[x] - 1/2*Log[b*x]^2 /;
FreeQ[b,x]
```

x: $\int \frac{\text{Gamma}[1,\,b\,x]}{x}\,dx$

Derivation: Algebraic expansion

Basis: $\text{Gamma}[1,\,z] == \frac{1}{e^z}$

Note: *Mathematica* automatically evaluates Gamma[1, z] to e^{-z}.

Rule: If $n > 1$, then

$$\int \frac{\text{Gamma}[1,\,b\,x]}{x}\,dx \longrightarrow \int \frac{1}{x\,e^{b\,x}}\,dx$$

Program code:

```
(* Int[Gamma[1,b_.*x_]/x_,x_Symbol] :=
  Int[1/(x*E^(b*x)),x] /;
FreeQ[b,x] *)
```

2: $\int \frac{\text{Gamma}[n, b\,x]}{x}\,dx$ when $n - 1 \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Basis: $\text{Gamma}[n, z] == \frac{z^{n-1}}{e^z} + (n-1)\,\text{Gamma}[n-1, z]$

Rule: If $n - 1 \in \mathbb{Z}^+$, then

$$\int \frac{\text{Gamma}[n, b\,x]}{x}\,dx \longrightarrow b \int \frac{(b\,x)^{n-2}}{e^{b\,x}}\,dx + (n-1) \int \frac{\text{Gamma}[n-1, b\,x]}{x}\,dx \longrightarrow -\text{Gamma}[n-1, b\,x] + (n-1) \int \frac{\text{Gamma}[n-1, b\,x]}{x}\,dx$$

Program code:

```
Int[Gamma[n_,b_.*x_]/x_,x_Symbol] :=
  -Gamma[n-1,b*x] + (n-1)*Int[Gamma[n-1,b*x]/x,x] /;
FreeQ[b,x] && IGtQ[n,1]
```

3: $\int \frac{\text{Gamma}[n, b\,x]}{x}\,dx$ when $n \in \mathbb{Z}^-$

Derivation: Algebraic expansion

Basis: $\text{Gamma}[n, z] == -\frac{z^n}{n\,e^z} + \frac{1}{n}\,\text{Gamma}[n+1, z]$

Rule: If $n \in \mathbb{Z}^-$, then

$$\int \frac{\text{Gamma}[n, b\,x]}{x}\,dx \longrightarrow -\frac{b}{n} \int \frac{(b\,x)^{n-1}}{e^{b\,x}}\,dx + \frac{1}{n} \int \frac{\text{Gamma}[n+1, b\,x]}{x}\,dx \longrightarrow \frac{\text{Gamma}[n, b\,x]}{n} + \frac{1}{n} \int \frac{\text{Gamma}[n+1, b\,x]}{x}\,dx$$

Program code:

```
Int[Gamma[n_,b_.*x_]/x_,x_Symbol] :=
  Gamma[n,b*x]/n + 1/n*Int[Gamma[n+1,b*x]/x,x] /;
FreeQ[b,x] && ILtQ[n,0]
```

2: $\int \frac{\text{Gamma}[n, b\,x]}{x}\,dx$ when $n \notin \mathbb{Z}$

Rule: If $n \notin \mathbb{Z}$, then

$$\int \frac{\text{Gamma}[n, b\,x]}{x}\,dx \longrightarrow \text{Gamma}[n]\,\text{Log}[x] - \frac{(b\,x)^n}{n^2}\,\text{HypergeometricPFQ}[\{n, n\}, \{1+n, 1+n\}, -b\,x]$$

Program code:

```
Int[Gamma[n_,b_.*x_]/x_,x_Symbol] :=
  Gamma[n]*Log[x] - (b*x)^n/n^2*HypergeometricPFQ[{n,n},{1+n,1+n},-b*x] /;
FreeQ[{b,n},x] && Not[IntegerQ[n]]
```

2: $\int (d\,x)^m \,\text{Gamma}[n,\, b\,x]\, dx$ when $m \neq -1$

Derivation: Integration by parts and piecewise constant extraction

- Basis: $\partial_x \frac{(d\,x)^m}{(b\,x)^m} == 0$

- Basis: $-\frac{1}{b}\,\partial_x \text{Gamma}[m+n+1,\, b\,x] == \frac{(b\,x)^{m+n}}{e^{b\,x}}$

- Note: The antiderivative is given directly without recursion so it is expressed entirely in terms of the incomplete gamma function without need for the exponential function.

- Rule: If $m \neq -1$, then

$$\int (d\,x)^m \,\text{Gamma}[n,\, b\,x]\, dx \longrightarrow \frac{(d\,x)^{m+1}\,\text{Gamma}[n,\, b\,x]}{d\,(m+1)} + \frac{1}{m+1}\int \frac{(d\,x)^m\,(b\,x)^n}{e^{b\,x}}\, dx$$
$$\longrightarrow \frac{(d\,x)^{m+1}\,\text{Gamma}[n,\, b\,x]}{d\,(m+1)} + \frac{(d\,x)^m}{(m+1)\,(b\,x)^m}\int \frac{(b\,x)^{m+n}}{e^{b\,x}}\, dx$$
$$\longrightarrow \frac{(d\,x)^{m+1}\,\text{Gamma}[n,\, b\,x]}{d\,(m+1)} - \frac{(d\,x)^m\,\text{Gamma}[m+n+1,\, b\,x]}{b\,(m+1)\,(b\,x)^m}$$

- Program code:

```
Int[(d_.*x_)^m_.*Gamma[n_,b_.*x_],x_Symbol] :=
  (d*x)^(m+1)*Gamma[n,b*x]/(d*(m+1)) -
  (d*x)^m*Gamma[m+n+1,b*x]/(b*(m+1)*(b*x)^m) /;
FreeQ[{b,d,m,n},x] && NeQ[m,-1]
```

3. $\int (c + d x)^m \, \text{Gamma}[n, a + b x] \, dx$

1: $\int (c + d x)^m \, \text{Gamma}[n, a + b x] \, dx$ when $b c - a d == 0$

Derivation: Integration by substitution

Rule: If $b c - a d == 0$, then

$$\int (c + d x)^m \, \text{Gamma}[n, a + b x] \, dx \longrightarrow \frac{1}{b} \text{Subst}\left[\int \left(\frac{d x}{b}\right)^m \text{Gamma}[n, x] \, dx, x, a + b x\right]$$

Program code:

```
Int[(c_+d_.*x_)^m_.*Gamma[n_,a_+b_.*x_],x_Symbol] :=
  1/b*Subst[Int[(d*x/b)^m*Gamma[n,x],x],x,a+b*x] /;
FreeQ[{a,b,c,d,m,n},x] && EqQ[b*c-a*d,0]
```

2: $\int \frac{\text{Gamma}[n, a + b x]}{c + d x} \, dx$ when $n - 1 \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Basis: $\text{Gamma}[n, z] == \frac{z^{n-1}}{e^z} + (n - 1) \, \text{Gamma}[n - 1, z]$

Rule: If $n - 1 \in \mathbb{Z}^+$, then

$$\int \frac{\text{Gamma}[n, a + b x]}{c + d x} \, dx \longrightarrow \int \frac{(a + b x)^{n-1}}{(c + d x) \, e^{a+b x}} \, dx + (n - 1) \int \frac{\text{Gamma}[n - 1, a + b x]}{c + d x} \, dx$$

Program code:

```
Int[Gamma[n_,a_.+b_.*x_]/(c_.+d_.*x_),x_Symbol] :=
  Int[(a+b*x)^(n-1)/((c+d*x)*E^(a+b*x)),x] + (n-1)*Int[Gamma[n-1,a+b*x]/(c+d*x),x] /;
FreeQ[{a,b,c,d},x] && IGtQ[n,1]
```

3: $\int (c+d x)^m \text{Gamma}[n, a+b x] \, dx$ when $(m \in \mathbb{Z}^+ \lor n \in \mathbb{Z}^+ \lor (m \mid n) \in \mathbb{Z}) \land m \neq -1$

Derivation: Integration by parts

Basis: $\partial_x \text{Gamma}[n, a+b x] == -\frac{b (a+b x)^{n-1}}{e^{a+b x}}$

Rule: If $(m \in \mathbb{Z}^+ \lor n \in \mathbb{Z}^+ \lor (m \mid n) \in \mathbb{Z}) \land m \neq -1$, then

$$\int (c+d x)^m \text{Gamma}[n, a+b x] \, dx \longrightarrow \frac{(c+d x)^{m+1} \text{Gamma}[n, a+b x]}{d (m+1)} + \frac{b}{d (m+1)} \int \frac{(c+d x)^{m+1} (a+b x)^{n-1}}{e^{a+b x}} \, dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*Gamma[n_,a_.+b_.*x_],x_Symbol] :=
  Block[{$UseGamma=True},
    (c+d*x)^(m+1)*Gamma[n,a+b*x]/(d*(m+1)) +
    b/(d*(m+1))*Int[(c+d*x)^(m+1)*(a+b*x)^(n-1)/E^(a+b*x),x]] /;
FreeQ[{a,b,c,d,m,n},x] && (IGtQ[m,0] || IGtQ[n,0] || IntegersQ[m,n]) && NeQ[m,-1]
```

U: $\int (c+d x)^m \text{Gamma}[n, a+b x] \, dx$

Rule:

$$\int (c+d x)^m \text{Gamma}[n, a+b x] \, dx \longrightarrow \int (c+d x)^m \text{Gamma}[n, a+b x] \, dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*Gamma[n_,a_.+b_.*x_],x_Symbol] :=
  Unintegrable[(c+d*x)^m*Gamma[n,a+b*x],x] /;
FreeQ[{a,b,c,d,m,n},x]
```

2. $\int u\,\text{LogGamma}[a+b\,x]\,dx$

1: $\int \text{LogGamma}[a+b\,x]\,dx$

Derivation: Primitive rule

Basis: $\frac{\partial \psi^{(-2)}(z)}{\partial z} = \log\Gamma(z)$

Rule:

$$\int \text{LogGamma}[a+b\,x]\,dx \longrightarrow \frac{\text{PolyGamma}[-2,\ a+b\,x]}{b}$$

–

Program code:

```
Int[LogGamma[a_.+b_.*x_],x_Symbol] :=
  PolyGamma[-2,a+b*x]/b /;
FreeQ[{a,b},x]
```

2. $\int (c+d\,x)^m\,\text{LogGamma}[a+b\,x]\,dx$

1: $\int (c+d\,x)^m\,\text{LogGamma}[a+b\,x]\,dx$ when $m\in\mathbb{Z}^+$

Derivation: Integration by parts

Rule: If $m\in\mathbb{Z}^+$, then

$$\int (c+d\,x)^m\,\text{LogGamma}[a+b\,x]\,dx \longrightarrow \frac{(c+d\,x)^m\,\text{PolyGamma}[-2,\,a+b\,x]}{b}-\frac{d\,m}{b}\int (c+d\,x)^{m-1}\,\text{PolyGamma}[-2,\,a+b\,x]\,dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*LogGamma[a_.+b_.*x_],x_Symbol] :=
  (c+d*x)^m*PolyGamma[-2,a+b*x]/b -
  d*m/b*Int[(c+d*x)^(m-1)*PolyGamma[-2,a+b*x],x] /;
FreeQ[{a,b,c,d},x] && IGtQ[m,0]
```

U: $\int (c+d\,x)^m\,\text{LogGamma}[a+b\,x]\,dx$

Rule:

$$\int (c+d\,x)^m\,\text{LogGamma}[a+b\,x]\,dx \longrightarrow \int (c+d\,x)^m\,\text{LogGamma}[a+b\,x]\,dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*LogGamma[a_.+b_.*x_],x_Symbol] :=
  Unintegrable[(c+d*x)^m*LogGamma[a+b*x],x] /;
FreeQ[{a,b,c,d,m},x]
```

3. $\int u \, \text{PolyGamma}[n, a + b x] \, dx$

1: $\int \text{PolyGamma}[n, a + b x] \, dx$

Derivation: Primitive rule

Basis: $\frac{\partial \psi^{(n)}(z)}{\partial z} = \psi^{(n+1)}(z)$

Rule:

$$\int \text{PolyGamma}[n, a + b x] \, dx \longrightarrow \frac{\text{PolyGamma}[n - 1, a + b x]}{b}$$

–

Program code:

```
Int[PolyGamma[n_,a_.+b_.*x_],x_Symbol] :=
  PolyGamma[n-1,a+b*x]/b /;
FreeQ[{a,b,n},x]
```

2. $\int (c+d\,x)^m \,\text{PolyGamma}[n,\,a+b\,x]\,dx$

1: $\int (c+d\,x)^m \,\text{PolyGamma}[n,\,a+b\,x]\,dx$ when $m>0$

Derivation: Integration by parts

Rule: If $m>0$, then

$$\int (c+d\,x)^m \,\text{PolyGamma}[n,\,a+b\,x]\,dx \longrightarrow \frac{(c+d\,x)^m \,\text{PolyGamma}[n-1,\,a+b\,x]}{b} - \frac{d\,m}{b}\int (c+d\,x)^{m-1}\,\text{PolyGamma}[n-1,\,a+b\,x]\,dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*PolyGamma[n_,a_.+b_.*x_],x_Symbol] :=
  (c+d*x)^m*PolyGamma[n-1,a+b*x]/b - d*m/b*Int[(c+d*x)^(m-1)*PolyGamma[n-1,a+b*x],x] /;
FreeQ[{a,b,c,d,n},x] && GtQ[m,0]
```

2: $\int (c+d\,x)^m \,\text{PolyGamma}[n,\,a+b\,x]\,dx$ when $m<-1$

Derivation: Inverted integration by parts

Rule: If $m<-1$, then

$$\int (c+d\,x)^m \,\text{PolyGamma}[n,\,a+b\,x]\,dx \longrightarrow \frac{(c+d\,x)^{m+1}\,\text{PolyGamma}[n,\,a+b\,x]}{d\,(m+1)} - \frac{b}{d\,(m+1)}\int (c+d\,x)^{m+1}\,\text{PolyGamma}[n+1,\,a+b\,x]\,dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*PolyGamma[n_,a_.+b_.*x_],x_Symbol] :=
  (c+d*x)^(m+1)*PolyGamma[n,a+b*x]/(d*(m+1)) -
  b/(d*(m+1))*Int[(c+d*x)^(m+1)*PolyGamma[n+1,a+b*x],x] /;
FreeQ[{a,b,c,d,n},x] && LtQ[m,-1]
```

U: $\int (c + d\,x)^m \,\text{PolyGamma}[n,\ a + b\,x]\,dx$

Rule:

$$\int (c + d\,x)^m \,\text{PolyGamma}[n,\ a + b\,x]\,dx \longrightarrow \int (c + d\,x)^m \,\text{PolyGamma}[n,\ a + b\,x]\,dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*PolyGamma[n_,a_.+b_.*x_],x_Symbol] :=
  Unintegrable[(c+d*x)^m*PolyGamma[n,a+b*x],x] /;
FreeQ[{a,b,c,d,m,n},x]
```

4: $\int \text{Gamma}[a + b\,x]^n \,\text{PolyGamma}[0,\ a + b\,x]\,dx$

Derivation: Primitive rule

Basis: $\frac{\partial \Gamma(z)^n}{\partial z} = n\,\psi^{(0)}(z)\,\Gamma(z)^n$

Rule:

$$\int \text{Gamma}[a + b\,x]^n \,\text{PolyGamma}[0,\ a + b\,x]\,dx \longrightarrow \frac{\text{Gamma}[a + b\,x]^n}{b\,n}$$

Program code:

```
Int[Gamma[a_.+b_.*x_]^n_.*PolyGamma[0,a_.+b_.*x_],x_Symbol] :=
  Gamma[a+b*x]^n/(b*n) /;
FreeQ[{a,b,n},x]
```

5: $\int ((a+b\,x)\,!)^n\,\text{PolyGamma}[0,\,c+b\,x]\,dx$ when $c == a+1$

Derivation: Primitive rule

Basis: $\frac{\partial (z!)^n}{\partial z} = n\,\psi^{(0)}(z+1)\,(z!)^n$

– Rule: If $c == a+1$, then

$$\int ((a+b\,x)\,!)^n\,\text{PolyGamma}[0,\,c+b\,x]\,dx \longrightarrow \frac{((a+b\,x)\,!)^n}{b\,n}$$

– Program code:

```
Int[((a_.+b_.*x_)!)^n_.*PolyGamma[0,c_.+b_.*x_],x_Symbol] :=
  ((a+b*x)!)^n/(b*n) /;
FreeQ[{a,b,c,n},x] && EqQ[c,a+1]
```

6. $\int u\,\text{Gamma}[p, d\,(a + b\,\text{Log}[c\,x^n])]\,dx$

1: $\int \text{Gamma}[p, d\,(a + b\,\text{Log}[c\,x^n])]\,dx$

Derivation: Integration by parts

Basis: $\partial_x \text{Gamma}[p, d\,(a + b\,\text{Log}[c\,x^n])] == -\frac{b\,d\,n\,e^{-a}\,(d\,(a+b\,\text{Log}[c\,x^n]))^{p-1}}{x\,(c\,x^n)^{b\,d}}$

Rule:

$$\int \text{Gamma}[p, d\,(a + b\,\text{Log}[c\,x^n])]\,dx \longrightarrow x\,\text{Gamma}[p, d\,(a + b\,\text{Log}[c\,x^n])] + b\,d\,n\,e^{-a\,d}\int \frac{(d\,(a + b\,\text{Log}[c\,x^n]))^{p-1}}{(c\,x^n)^{b\,d}}\,dx$$

Program code:

```
Int[Gamma[p_,d_.*(a_.+b_.*Log[c_.*x_^n_.])],x_Symbol] :=
  x*Gamma[p,d*(a+b*Log[c*x^n])] + b*d*n*E^(-a*d)*Int[(d*(a+b*Log[c*x^n]))^(p-1)/(c*x^n)^(b*d),x] /;
FreeQ[{a,b,c,d,n,p},x]
```

2: $\int \frac{\text{Gamma}[p, d\,(a + b\,\text{Log}[c\,x^n])]}{x}\,dx$

Derivation: Integration by substitution

Basis: $\frac{F[\text{Log}[c\,x^n]]}{x} == \frac{1}{n}\,\text{Subst}[F[x], x, \text{Log}[c\,x^n]]\,\partial_x \text{Log}[c\,x^n]$

Rule:

$$\int \frac{\text{Gamma}[p, d\,(a + b\,\text{Log}[c\,x^n])]}{x}\,dx \longrightarrow \frac{1}{n}\,\text{Subst}[\text{Gamma}[p, d\,(a + b\,x)], x, \text{Log}[c\,x^n]]$$

Program code:

```
Int[Gamma[p_,d_.*(a_.+b_.*Log[c_.*x_^n_.])]/x_,x_Symbol] :=
  1/n*Subst[Gamma[p,d*(a+b*x)],x,Log[c*x^n]] /;
FreeQ[{a,b,c,d,n,p},x]
```

3: $\int (e\,x)^m\,\text{Gamma}\left[p,\ d\left(a+b\,\text{Log}\left[c\,x^n\right]\right)\right]\,dx$ when $m\neq -1$

Derivation: Integration by parts

Basis: $\partial_x\,\text{Gamma}\left[p,\ d\left(a+b\,\text{Log}\left[c\,x^n\right]\right)\right]==-\frac{b\,d\,n\,e^{-a\,d}\left(d\left(a+b\,\text{Log}\left[c\,x^n\right]\right)\right)^{-1+p}}{x\left(c\,x^n\right)^{b\,d}}$

Rule: If $m\neq -1$, then

$$\int (e\,x)^m\,\text{Gamma}\left[p,\ d\left(a+b\,\text{Log}\left[c\,x^n\right]\right)\right]\,dx\ \longrightarrow\ \frac{(e\,x)^{m+1}\,\text{Gamma}\left[p,\ d\left(a+b\,\text{Log}\left[c\,x^n\right]\right)\right]}{e\,(m+1)}+\frac{b\,d\,n\,e^{-a\,d}\,(e\,x)^{b\,d\,n}}{(m+1)\left(c\,x^n\right)^{b\,d}}\int (e\,x)^{m-b\,d\,n}\left(d\left(a+b\,\text{Log}\left[c\,x^n\right]\right)\right)^{p-1}\,dx$$

Program code:

```
Int[(e_.*x_)^m_.*Gamma[p_,d_.*(a_.+b_.*Log[c_.*x_^n_.])],x_Symbol] :=
  (e*x)^(m+1)*Gamma[p,d*(a+b*Log[c*x^n])]/(e*(m+1)) +
  b*d*n*E^(-a*d)*(e*x)^(b*d*n)/((m+1)*(c*x^n)^(b*d))*Int[(e*x)^(m-b*d*n)*(d*(a+b*Log[c*x^n]))^(p-1),x] /;
FreeQ[{a,b,c,d,e,m,n,p},x] && NeQ[m,-1]
```

7. $\int u\, \text{Gamma}\left[p,\ f\left(a+b\, \text{Log}\left[c\, (d+e\, x)^n\right]\right)\right] dx$

1: $\int \text{Gamma}\left[p,\ f\left(a+b\, \text{Log}\left[c\, (d+e\, x)^n\right]\right)\right] dx$

Derivation: Integration by substitution

– Rule:

$$\int \text{Gamma}\left[p,\ f\left(a+b\, \text{Log}\left[c\, (d+e\, x)^n\right]\right)\right] dx \longrightarrow \frac{1}{e}\, \text{Subst}\left[\int \text{Gamma}\left[p,\ f\left(a+b\, \text{Log}\left[c\, x^n\right]\right)\right] dx,\ x,\ d+e\, x\right]$$

– Program code:

```
Int[Gamma[p_,f_.*(a_.+b_.*Log[c_.*(d_+e_.*x_)^n_.])],x_Symbol] :=
  1/e*Subst[Int[Gamma[p,f*(a+b*Log[c*x^n])],x],x,d+e*x] /;
FreeQ[{a,b,c,d,e,f,n,p},x]
```

2: $\int (g+h\, x)^m\, \text{Gamma}\left[p,\ f\left(a+b\, \text{Log}\left[c\, (d+e\, x)^n\right]\right)\right] dx$ when $e\, g-d\, h=0$

Derivation: Integration by substitution

Basis: If $e\, g-d\, h=0$, then $(g+h\, x)^m\, F[d+e\, x] = \frac{1}{e}\, \text{Subst}\left[\left(\frac{g\, x}{d}\right)^m F[x],\ x,\ d+e\, x\right] \partial_x (d+e\, x)$

– Rule: If $e\, g-d\, h=0$, then

$$\int (g+h\, x)^m\, \text{Gamma}\left[p,\ f\left(a+b\, \text{Log}\left[c\, (d+e\, x)^n\right]\right)\right] dx \longrightarrow \frac{1}{e}\, \text{Subst}\left[\int \left(\frac{g\, x}{d}\right)^m \text{Gamma}\left[p,\ f\left(a+b\, \text{Log}\left[c\, x^n\right]\right)\right] dx,\ x,\ d+e\, x\right]$$

– Program code:

```
Int[(g_.+h_.*x_)^m_.*Gamma[p_,f_.*(a_.+b_.*Log[c_.*(d_+e_.*x_)^n_.])],x_Symbol] :=
  1/e*Subst[Int[(g*x/d)^m*Gamma[p,f*(a+b*Log[c*x^n])],x],x,d+e*x] /;
FreeQ[{a,b,c,d,e,f,g,h,m,n,p},x] && EqQ[e*g-d*h,0]
```